

# Calculating RRKM Rate Constants from Vibrational Frequencies and Their Dynamic Interpretation

Adam C. Mansell  
David J. Kahle  
Darrin J. Bellert

Rice–Rampsberger–Kassel–Marcus (RRKM) theory calculates an energy-dependent microcanonical unimolecular rate constant for a chemical reaction from a sum and density of vibrational quantum states. This article demonstrates how to program the Beyer–Swinehart direct count of the sum and density of states for harmonic oscillators, as well as the Stein–Rabinovitch extension for anharmonic oscillators. Microcanonical rate constants are calculated for the decomposition of vinyl cyanide ( $\text{C}_3\text{H}_3\text{N}$ ) into HCN, HNC and HCCH as an example.

## ■ 1. Introduction

The essential framework for our understanding of unimolecular reaction rates was developed largely in the first half of the twentieth century [1]. The *Lindemann–Hinshelwood mechanism*, proposed in 1922 by Lindemann and later expanded upon by Hinshelwood [2], posited that the gas phase unimolecular reactions were initiated by bimolecular collisions. This allowed the rate of the reaction to be calculated as a function of internal energy, regardless of the method of activation. Later that decade, Rice and Ramsperger and, independently, Kassel developed an improvement on the model known as the *Rice–Ramsperger–Kassel (RRK)* model [3, 4]. The RRK model viewed the molecule as a system of identical harmonic oscillators and introduced the idea of activation energy: that sufficient energy must be deposited into specific modes of motion in order for the reaction to occur. In 1952, Rice and Marcus extended this into what is now known as *Rice–Rampsberger–Kassel–Marcus* theory (RRKM), which incorporates a more complete quantum mechanical description of the molecule and utilizes the concept of the transition state, a specific conformation that the molecule must adopt in order for the

reaction to proceed [5]. The RRKM model remains the prevailing scientific description of unimolecular chemical kinetics to this day.

Chemical transformations require bonds to break in the reactants and new bonds to form as products are generated. The potential energy associated with these transformations defines an  $n$ -dimensional *potential energy surface* (PES), where  $n$  is the number of atoms [6]. The *reaction pathway* is a one-dimensional cross section of the PES composed of the precursor, intermediates, transition states and products of a particular reaction [6]. Reactants and intermediates of a reaction correspond to minima, whereas transition states are located at maxima along the reaction pathway.

The rate of the reaction depends on the probability of the molecule adopting the conformations along the reaction pathway, which, in turn, stems from the probability that a sufficient amount of energy is partitioned into the necessary modes of motion. The RRKM formalism assumes that: (1) once the molecule adopts the transition state orientation, the reaction proceeds to products; and (2) the *internal vibrational energy redistribution* (IVR) is fast with respect to the timescale of the reaction. Under these constraints, the RRKM microcanonical rate constant,  $k(E)$ , is given by

$$k(E) = \frac{\sigma N^\ddagger(E - E_0)}{h \rho(E)}, \quad (1)$$

where  $E$  is the internal energy of the system,  $E_0$  is the activation energy for the reaction,  $N^\ddagger(E - E_0)$  is the sum of states of the transition state from  $E_0$  to  $E$ ,  $\rho(E)$  is the density of states of the precursor at energy  $E$ ,  $h$  is Planck's constant, and  $\sigma$  is the degeneracy of the reaction pathway [3, 4, 5]. A state is defined as any unique vibrational configuration that determines the internal energy of the molecule. The sum of states,  $N(E)$ , is the total number of states within a specific energy range [6]. The density of states is the number of states per energy level. Equivalently, the density of states is the derivative of the sum of states with respect to energy [6]. The density is expressed in units of inverse energy ( $E^{-1}$ ), while the sum of states is a dimensionless quantity. Thus, the microcanonical rate constant of equation (1) has units  $s^{-1}$ .

There are  $3n - 6$  vibrational frequencies for nonlinear polyatomic molecules and  $3n - 7$  vibrational frequencies for the corresponding transition state. It is the values of these frequencies, which typically range from  $50$ – $3500\text{ cm}^{-1}$ , that dictate how the internal energy is distributed in the molecule. The calculation for the sum and density of states depends on these frequencies, which can be measured or looked up in standard data tables. However, for transition states, these values must be estimated or, more likely, calculated using theoretical packages.


This article is divided into the following sections: In Section 2, the sum and density of states for vinyl cyanide ( $C_3H_3N$ ) are determined through a direct count method analogous to that developed by Beyer and Swinehart [7]. In Section 3, the method is extended to anharmonic oscillators using the Stein–Rabinovitch algorithm [8]. In Section 4, the RRKM microcanonical rate constants are calculated for the unimolecular dissociation of vinyl cyanide and are used to predict various dynamic properties of the chemical reaction.

## ■ 2. Direct Count for Harmonic Oscillators: The Beyer–Swinehart Algorithm

*Beyer and Swinehart (BS)* developed a surprisingly simple yet computationally intensive direct count method to calculate the sum and density of states at defined energies of a harmonic oscillator [7]. The advantage of conducting such calculations in Mathematica is that the user can leverage high-quality numerical ODE solvers and interactive graphical features to visualize dynamic properties of the chemical reaction. This is demonstrated in Section 4, where the temporal development of nine different species is plotted and automatically updated as the user sets the internal energy of the system with a slider. All of this can be done in real time after some initial computations that are exact with Mathematica.

There are a number of equivalent ways to implement the BS direct count of states. To ease the exposition, we first illustrate the task with a `Do` loop. The sum and density of states are determined by the vibrational frequencies of the species of interest, which we define in the list `vList`. In more complex situations, such as biomolecules or reactions at very high energies, to lessen the computational burden it may be necessary to divide the value of the total energy into larger packets and modify the vibrational frequencies to be multiples of the new packet size [8]. In this case, the packet size is  $1\text{ cm}^{-1}$  and the vibrational frequencies for vinyl cyanide are taken from reference [9].

```
ChemicalData["VinylCyanide", "MoleculePlot"]
```



```
vList = {235, 344, 570, 696, 872, 980, 1005, 1118, 1331,  
        1467, 1689, 2322, 3173, 3214, 3271};  
nFreqs = Length[vList];
```

The final result for the sum or density of states is a list with  $energy + 1$  elements. For the sum of states, this list is initialized with all values set to 1. The sum of states can then be calculated using a nested (or doubly indexed) Do loop.

```
energy = 100 000;

sumOfStates = Table[1, energy + 1];
Do[
  Do[
    sumOfStates[[w + 1]] += sumOfStates[[w + 1 - vList[[j]]]],
    {w, vList[[j]], energy}
  ],
  {j, nFreqs}
];
```


Because the density of states is the derivative of the sum of states, the density of states can also be determined from the sum of states table by setting the  $i^{\text{th}}$  element of the density of states to the difference between the  $i^{\text{th}}$  and  $(i - 1)^{\text{th}}$  elements of the sum of states table. Alternatively and equivalently, this can be done to the initial sum of states table to derive the initial density of states table, which is a list with  $energy + 1$  elements with the initial element set to 1 and all other elements set to 0. This can be conveniently done with PadRight.

```
densityOfStates = PadRight[{1}, energy + 1];
Do[
  Do[
    densityOfStates[[w + 1]] +=
    densityOfStates[[w + 1 - vList[[j]]]],
    {w, vList[[j]], energy}
  ],
  {j, nFreqs}
];
```

A visual comparison of the sum and density of states, calculated for vinyl cyanide ( $\text{C}_3\text{H}_3\text{N}$ ), shows the behavior of the two quantities over the  $0 - 100\,000\text{ cm}^{-1}$  energy range. As the general trend of the graphic is clear, we subsample the lists of values to avoid plotting millions of points, needlessly increasing file size.

```
takeEvery[nth_] :=
Function[list,
  list[[Round@Subdivide[1, Length[list], nth]]]]
```

```
Show[
  ListLogPlot[takeEvery[2000]@sumOfStates, PlotStyle -> Red],
  ListLogPlot[takeEvery[2000]@densityOfStates],
  Frame -> True,
  FrameLabel -> {"Energy (cm-1)",
    "Sum of States (unitless)\n and Density of States\n (/cm-1)"}
]
```



Thus, when vinyl cyanide has an internal energy of  $100\,000\text{ cm}^{-1}$ , the sum and density of quantum states are approximately  $10^{17.5}$  and  $10^{14}$  / wavenumber.

Notice that the DO loops for computing the sum and density of states perform the same basic process—the BS direct count method—on different initial lists. As we proceed, adopting a functional programming approach eases the exposition and vastly improves code clarity. We do this by introducing two new functions, `bsFrequencyUpdate` and `bsDirectCount`. The first updates the lists of sums and densities for a single vibrational frequency, and the second uses `Fold` to iteratively update for each of the vibrational frequencies in the table.

```
bsFrequencyUpdate[stateList_, v_] :=
  Block[{out = stateList, k},
    For[k = v + 1, k ≤ Length[out], k++, out[[k]] += out[[k - v]]];
    out
  ]

bsDirectCount[stateList_, vList_] :=
  Fold[bsFrequencyUpdate, stateList, vList]
```

We then use these to create the two functions `bsComputeSumOfStates` and `bsComputeDensityOfStates`, which take a frequency list and total energy as arguments and do just as their names imply. The last argument to these functions, `relEnergy`, is used in Section 4.

```
bsComputeSumOfStates[vList_, energy_, relEnergy_: 0] :=  
  bsDirectCount[Table[1, energy - relEnergy + 1], vList]  
  
bsComputeDensityOfStates[vList_, energy_, relEnergy_: 0] :=  
  bsDirectCount[PadRight[{1}, energy - relEnergy + 1], vList]
```

For completeness, we recompute the sum and density of states using these functions to illustrate how that can be done. It is readily verified that these compute the same quantities.

```
bsSumOfStates = bsComputeSumOfStates[vList, energy];  
bsDensityOfStates = bsComputeDensityOfStates[vList, energy];
```

### ■ 3. Direct Count for Anharmonic Oscillators: The Stein–Rabinovitch Extension

The harmonic oscillator model assumes that the difference between successive vibrational energy levels remains constant. In real oscillators, this difference decreases as the vibrational excitation increases. These anharmonic effects increase both the sum and density of states and thus often cancel when taking the ratio to determine the RRKM rate constant in equation (1). However, there are some cases where this effect does not completely cancel or where highly resolved measurements necessitate a more accurate description of the vibrational energy.

The *Stein–Rabinovitch* (SR) algorithm extends the BS method to incorporate the effects of anharmonicity into the RRKM rate constant [8]. Following [10], the anharmonic effects are incorporated into the energy level expression through the equation

$$E(n) = v_e \left( n + \frac{1}{2} \right) - v_e x_e \left( n + \frac{1}{2} \right)^2, \quad (2)$$

$$n = 0, 1, 2, \dots,$$

where  $E(n)$  is the energy of the  $n^{\text{th}}$  vibrational level,  $v_e$  is the vibrational frequency, and  $x_e$  is the anharmonicity constant. The subscript e stands for equilibrium. It is chosen because this would be the vibrational frequency if the oscillator vibrated harmonically about its equilibrium position (the minimum in the well). The SR extension counts each energy level of the anharmonic model to calculate the sum and density of states in a way that is functionally similar to the BS method; however, the input table is modified to include the anharmonicity constant for each vibrational mode. For this example, we have organized the anharmonicity constants,  $x_i$ , into the table `xList`. Using the vibrational frequencies of vinyl cyanide from [9], and assigning 0.01 as the anharmonicity constant for each vibration, the energy for each level of each mode can be calculated.

```

xList = Table[1 / 100, {i, nFreqs}];

vEnergies[vList_, xList_] :=
Block[{nFreqs = Length[vList], i, n},
Round@Table[
{v, x} = {vList[[i]], xList[[i]]};
v + v n (1 - n x),
{i, nFreqs}, {n, 0,  $\frac{1}{2} \left( \frac{1}{xList[[i]]} - 1 \right)}$ }
]
]

```

With these we can calculate the sum and density of states, initializing the lists in the same manner as the BS method (a list of 1s for the sum of states, and a 1 followed by 0s for the density of states). The process is functionally similar to that used to calculate the sum and density of states for a set of harmonic oscillators. In the BS method, the elements of a single table are modified with each iteration of the inner loop. By contrast, in the SR extension, the elements of the table are updated through separable computations combined after each complete loop for each oscillator.

```

srShift[stateList_, v_] :=
Table[0, v] ~Join~ Drop[stateList, -v]

srShiftListTotal[stateList_, vList_] :=
Total@ParallelMap[srShift[stateList, #] &,
Prepend[vList, 0]]

srCount[stateList_, listOfvLists_] :=
Fold[srShiftListTotal, stateList, listOfvLists]

```

The implementations can be verified by replicating the calculations in the appendix of [8], which shows the calculation of the density of states up to 10 units of energy for a model system with two oscillators. The energies of the first oscillator are {2, 2, 4, 7} and the energies of the second are {0, 0, 2, 5}.

```

FoldList[
srShiftListTotal,
{1, 0, 0, 0, 0, 0, 0, 0, 0, 0},
{{2, 2, 4, 7}, {0, 0, 2, 5}}
] // TableForm

```

1	0	0	0	0	0	0	0	0	0	0
1	0	2	0	1	0	0	1	0	0	0
3	0	7	0	5	1	1	5	0	2	0

With these, we can define SR analogs to `bsComputeSumOfStates` and `bsComputeDensityOfStates`.

```
srComputeSumOfStates[vList_, xList_, energy_,
  relEnergy_ : 0] :=
  srCount[Table[1, energy - relEnergy + 1],
    vEnergies[vList, xList]]


srComputeDensitiesOfStates[vList_, xList_, energy_,
  relEnergy_ : 0] :=
  srCount[PadRight[{1}, energy - relEnergy + 1],
    vEnergies[vList, xList]]
```

We can then use them to compute the SR sum and density of states.

```
srSumOfStates = srComputeDensitiesOfStates[vList,
  xList, energy];
srDensityOfStates = srComputeDensitiesOfStates[vList,
  xList, energy];
```


Here are two plots comparing the harmonic and anharmonic models for the density and sum of states calculated for vinyl cyanide.

```
Show[
  ListLogPlot[takeEvery[2000]@bsDensityOfStates,
    PlotStyle -> Red],
  ListLogPlot[takeEvery[2000]@srDensityOfStates],
  Frame -> True,
  FrameLabel -> {"Energy (cm-1)", "Density of States (/cm-1)"}
]
```












```
Show[
  ListLogPlot[takeEvery[2000]@bsSumOfStates,
    PlotStyle -> Red],
  ListLogPlot[takeEvery[2000]@srSumOfStates],
  Frame -> True,
  FrameLabel -> {"Energy (cm-1)", "Sum of States (unitless)"}
]
```



#### ■ 4. Dissociation of Vinyl Cyanide C<sub>3</sub> H<sub>3</sub> N

As a more substantive example, we calculate rate constants for the decomposition of vinyl cyanide VC (CH<sub>2</sub> CHCN) into hydrogen cyanide (HCN), hydrogen isocyanide (HNC) and acetylene (C<sub>2</sub> H<sub>2</sub>), compare them to published values, and interactively visualize the temporal dependence of the various species. In [9], Homayoon and colleagues considered the following kinetic model that involved seven different paths, three different intermediates, 13 rate constants and three different products:





Each step of each path passes through a transition state to yield either an intermediate or a final product. The following differential rate equations characterize the temporal dependence of the amounts of each species:

$$\frac{d}{dt} \text{VC} = -(k_1 + k_2 + k_3 + k_6 + k_9) \text{VC}(t) + k_4 \text{Int1}_{\text{III}}(t) + k_7 \text{Int1}_{\text{IV}}(t) + k_{10} \text{Int1}_V(t),$$

$$\frac{d}{dt} \text{Int1}_{\text{III}} = k_3 \text{VC}(t) - k_4 \text{Int1}_{\text{III}}(t) - k_5 \text{Int1}_{\text{III}}(t) - k_{12} \text{Int1}_{\text{III}}(t) - k_{13} \text{Int1}_{\text{III}}(t),$$

$$\frac{d}{dt} \text{Int1}_{\text{IV}} = k_6 \text{VC}(t) - k_7 \text{Int1}_{\text{IV}}(t) - k_8 \text{Int1}_{\text{IV}}(t),$$

$$\frac{d}{dt} \text{Int1}_V = k_9 \text{VC}(t) - k_{10} \text{Int1}_V(t) - k_{11} \text{Int1}_V(t),$$

$$\frac{d}{dt} \text{HCN} = (k_1 + k_2) \text{VC}(t) + k_5 \text{Int1}_{\text{III}}(t),$$

$$\frac{d}{dt} \text{HNC} = k_8 \text{Int1}_{\text{IV}}(t) + k_{11} \text{Int1}_V(t) + k_{12} \text{Int1}_{\text{III}}(t) + k_{13} \text{Int1}_{\text{III}}(t),$$

$$\begin{aligned} \frac{d}{dt} \text{HCCH} = & k_2 \text{VC}(t) + k_5 \text{Int1}_{\text{III}}(t) + k_8 \text{Int1}_{\text{IV}}(t) + \\ & k_{13} \text{Int1}_{\text{III}}(t) + k_1 \text{VC}(t) + k_{11} \text{Int1}_V(t) + k_{12} \text{Int1}_{\text{III}}(t). \end{aligned}$$

The determination of the sum and density of states requires the vibrational frequencies for each species located at minima and maxima along the reaction pathway. For the decomposition reaction of vinyl cyanide, there are six minima (the geometry of VC along with each intermediate) and 11 maxima (each unique transition state). These 17 sets of frequencies are taken from reference [9] and are provided in Table 1. In the following definitions, we have chosen the same notation as in reference [9]. TS indicates a transition state, the particular path is indicated by a Roman numeral, and the minimum energy of each species relative to the minimum energy of vinyl cyanide has the prefix *re*.

Species	Frequencies (cm <sup>-1</sup> )
VC	235, 344, 570, 696, 872, 980, 1005, 1118, 1331, 1467, 1689, 2322, 3173, 3214, 3271
TS1 – I	127, 184, 276, 521, 596, 778, 898, 953, 1353, 1615, 2187, 2272, 3153, 3271
TS1 – II	49, 106, 301, 384, 627, 696, 760, 790, 907, 1853, 2085, 2168, 3353, 3427
TS1 – III	237, 413, 604, 685, 958, 1013, 1044, 1303, 1415, 1652, 1979, 3175, 3264, 3280
Int1 – III	192, 237, 520, 707, 893, 942, 988, 1141, 1345, 1458, 1698, 2208, 3182, 3230, 3284
TS2 – III	124, 253, 338, 405, 600, 678, 891, 900, 976, 1759, 1877, 2123, 3303, 3403
TS1 – IV	411, 469, 570, 597, 753, 954, 970, 1076, 1272, 1365, 1778, 2120, 3047, 3211
INT1 – IV	452, 457, 713, 844, 845, 890, 930, 1048, 1146, 1281, 1576, 1832, 3245, 3281, 3516
TS2 – IV	346, 392, 503, 633, 715, 917, 989, 1088, 1278, 1444, 2113, 3050, 3124, 3718
INT2 – IV	229, 344, 516, 591, 670, 739, 941, 990, 1156, 1254, 1414, 2122, 2981, 3116, 3692
TS3 – IV	78, 282, 525, 630, 632, 687, 871, 882, 1156, 1708, 1837, 3320, 3364, 3625
TS1 – V	208, 255, 428, 584, 718, 881, 972, 1081, 1481, 1686, 1974, 2338, 3126, 3220
Int1 – V	162, 221, 379, 589, 866, 904, 953, 1073, 1147, 1489, 1707, 2206, 3132, 3214, 3498
TS2 – V	84, 115, 272, 401, 436, 474, 664, 803, 1289, 1651, 2113, 3128, 3284, 3839
INT2 – V	74, 76, 148, 157, 164, 436, 743, 772, 779, 1258, 1704, 2093, 3137, 3230, 3579
TS1 – VI	111, 150, 210, 447, 466, 835, 853, 920, 1308, 1624, 2053, 2176, 3143, 3260
TS1 – VII	40, 132, 330, 395, 615, 682, 855, 877, 935, 1813, 2060, 2097, 3329, 3409

▲ **Table 1.** Vibrational frequencies for vinyl cyanide and each intermediate and transition state (each with one imaginary frequency) in the decomposition reaction. Values taken from [9].

We now turn to the calculation of the sum and density of states and then use the values to calculate each rate constant. We begin by defining the vibrational constants.

```

vVC = {235, 344, 570, 696, 872, 980, 1005, 1118, 1331,
       1467, 1689, 2322, 3173, 3214, 3271};

```

```

(* Path I *)

```

```

vTS1_I = {127, 184, 276, 521, 596, 778, 898, 953, 1353,
          1615, 2187, 2272, 3153, 3271};

```

```

(* Path II *)

```

```

vTS1_II = {49, 106, 301, 384, 627, 696, 760, 790, 907,
           1853, 2085, 2168, 3353, 3427};

```

```

(* Path III *)

```

```

vTS1_III = {237, 413, 604, 685, 958, 1013, 1044, 1303,
            1415, 1652, 1979, 3175, 3264, 3280};
vInt1_III = {192, 237, 520, 707, 893, 942, 988, 1141, 1345,
             1458, 1698, 2208, 3182, 3230, 3284};
vTS2_III = {124, 253, 338, 405, 600, 678, 891, 900, 976,
            1759, 1877, 2123, 3303, 3403};

```

```

(* Path IV *)

```

```

vTS1_IV = {411, 469, 570, 597, 753, 954, 970, 1076, 1272,
           1365, 1778, 2120, 3047, 3211};
vInt1_IV = {452, 457, 713, 844, 845, 890, 930, 1048, 1146,
            1281, 1576, 1832, 3245, 3281, 3516};
vTS2_IV = {78, 282, 525, 630, 632, 687, 871, 882, 1156,
            1708, 1837, 3320, 3364, 3625};

```

```

(* Path V *)

```

```

vTS1_V = {208, 255, 428, 584, 718, 881, 972, 1081, 1481,
          1686, 1974, 2338, 3126, 3220};
vInt1_V = {162, 221, 379, 589, 866, 904, 953, 1073, 1147,
            1489, 1707, 2206, 3132, 3214, 3498};
vTS2_V = {84, 115, 272, 401, 436, 474, 664, 803, 1289,
           1651, 2113, 3128, 3284, 3839};

```

```

(* Path VI *)

```

```

vTS1_VI = {111, 150, 210, 447, 466, 835, 853, 920, 1308,
           1624, 2053, 2176, 3143, 3260};

```

```

(* Path VII *)

```

```

vTS1_VII = {40, 132, 330, 395, 615, 682, 855, 877, 935,
            1813, 2060, 2097, 3329, 3409};

```

Because the internal energy is defined relative to the well depth of vinyl cyanide, the sums and densities of states for the other species must be calculated using energy adjusted by the lowest energy of each species relative to the lowest energy of vinyl cyanide (peak height for the transition states and well depth for the intermediates).

```
reVC = 0;
reTS1_I = 35 185;
reTS1_II = 41 271;
reTS1_III = 20 076;
reInt1_III = 7380;
reTS2_III = 40 327;
reTS1_IV = 37 494;
reInt1_IV = 17 418;
reTS2_IV = 29 939;
reTS1_V = 36 724;
reInt1_V = 13 675;
reTS2_V = 33 402;
reTS1_VI = 38 508;
reTS1_VII = 40 292;
```

#### □ Beyer–Swinehart Sum and Density of States

Using `bsComputeSumOfStates` and `bsComputeDensityOfStates` from Section 1, we can now compute the sums and densities of states in parallel.

```
sumsOfStates =
Parallelize@
MapThread[bsComputeSumOfStates[#1, energy, #2] &, {
  {vTS1_I, vTS1_II, vTS1_III, vTS2_III, vTS1_IV, vTS2_IV,
   vTS1_V, vTS2_V, vTS1_VI, vTS1_VII},
  {reTS1_I, reTS1_II, reTS1_III, reTS2_III, reTS1_IV,
   reTS2_IV, reTS1_V, reTS2_V, reTS1_VI, reTS1_VII}
}];

densitiesOfStates =
Parallelize@
MapThread[bsComputeDensityOfStates[#1, energy, #2] &, {
  {vVC, vInt1_III, vInt1_IV, vInt1_V},
  {reVC, reInt1_III, reInt1_IV, reInt1_V}
}];
```

## □ Rate Constants

Using equation (1), we now compute the 13 rate constants, taking care to avoid indeterminate values such as 0/0.

```
h =
  QuantityMagnitude@
    UnitConvert[Quantity[1, "PlanckConstant"], "SIBase"] *
    5.034 * 1022;
(* Multiplying by 5.034*1022 incorporates the
   conversion of Wavenumbers to Joules into the rate
   constant determination step. *)


ple[vec_] := PadLeft[vec, energy + 1];
(* PadLeft to same length as energy *)

paddedDivide[x_, y_] :=
  Quiet[ple[x] / ple[y]] /. {ComplexInfinity → 0}

{k1, k2, k3, k4, k5, k6, k7, k8, k9, k10, k11, k12, k13} =
  MapThread[paddedDivide, {
    sumsOfStates[{{1, 2, 3, 3, 4, 5, 5, 6, 7, 7, 8, 9, 10}}],
    h densitiesOfStates[{{1, 1, 1, 2, 2, 1, 3, 3, 1, 4,
      4, 2, 2}}]
  }];
```

The rate constants can now be visualized. Note that as energy increases, all the reaction rates also increase.

```
Show[
  ListLinePlot@Log10[takeEvery[2000] /@ {k1, k2}],
  ImageSize → 400, Frame → True,
  FrameLabel → {"Energy (cm-1)", "k(E) (s-1)"},
  FrameTicks →
    {Automatic, {#, HoldForm[10#]} & /@ Range[0, 40, 2]}
]
```



## □ Temporal Dependence of the Various Species

The temporal dependence of the relative concentrations of each stable species (precursor, intermediates and products) can be determined using the rate equations and `NDSolve`. The reaction dynamics of the decay of the precursor vinyl cyanide, the subsequent build up and decay of the intermediates, and the formation of products, each normalized with respect to VC, are plotted at a given internal energy determined by a variable slide bar.

```
Manipulate[
  Plot[
    Evaluate[
      {VC[t], Int1_III[t], Int1_IV[t], Int1_V[t], HCN[t],
        HNC[t], HCCH[t]} /.
      NDSolve[{
        
$$\frac{VC'[t]}{h} = -(k1[p] + k2[p] + k3[p] + k6[p] + k9[p])$$

        VC[t] + k4[p] Int1_III[t] + k7[p] Int1_IV[t] +
        k10[p] Int1_V[t],
        
$$h^{-1} Int1_{III}'[t] = k3[p] VC[t] -$$

        (k4[p] + k5[p] + k12[p] + k13[p]) Int1_III[t],
        
$$h^{-1} Int1_{IV}'[t] = k6[p] VC[t] -$$

        (k7[p] + k8[p]) Int1_IV[t],
        
$$h^{-1} Int1_V'[t] = k9[p] VC[t] -$$

        (k10[p] + k11[p]) Int1_V[t],
        
$$h^{-1} HCN'[t] = (k1[p] + k2[p]) VC[t] +$$

        k5[p] Int1_III[t],
        
$$h^{-1} HNC'[t] = k8[p] Int1_IV[t] + k11[p] Int1_V[t] +$$

        (k12[p] + k13[p]) Int1_III[t],
        
$$h^{-1} HCCH'[t] = (k1[p] + k2[p]) VC[t] +$$


        (k5[p] + k12[p] + k13[p]) Int1_III[t] +
        k8[p] Int1_IV[t] + k11[p] Int1_V[t],

        VC[0] == 1, Int1_III[0] == 0, Int1_IV[0] == 0, Int1_V[0] == 0,
        HCN[0] == 0, HNC[0] == 0, HCCH[0] == 0
      }, {VC, Int1_III, Int1_IV, Int1_V, HCN, HNC, HCCH},
      {t, 0, 2}
    ] (* End NDSolve *)
  ], (* End Evaluate *)
  {t, 0, 2}, PlotRange -> {{0, 1.5}, {-.05, 1.05}},
  Frame -> True,
  FrameStyle -> Directive[12, Thickness[.0025]],
  PlotStyle -> {{Red}, {Green}, {Blue}, {Black}, {Gray},
    {Cyan}, {Magenta}},
  FrameLabel -> {Text@Style["Time (10-13 s)", 12],
```

```

Text@Style["Relative Abundance", 12]},
PlotLegends →
LineLegend[{Red, Green, Blue, Black, Gray, Cyan, Magenta},
{"VC", "Int1III", "Int1IV", "Int1V", "HCN", "HNC",
"HCCH"}]
], (* End Plot *)
{{p, 62 500, Text@Style["Energy (cm-1)", 12]}, 21 000,
100 000, 100},
Paneled → False, SaveDefinitions → True
]

```



A number of qualitative and quantitative results can be gleaned by simple inspection of this dynamic plot. One example is the amount of time it takes for the reaction to run to completion as energy increases. The plot also indicates that there is never a significant amount of any intermediate apart from that in path III. This is not surprising, given the significantly lower energy for the first transition state in path III relative to the first transition state in pathways IV and V, and given that three pathways share Int1<sub>III</sub>.



A final element determined from this analysis is the exit channel ratios (or product distribution ratios) determined at reaction completion and for a specific system internal energy. This is important because many experimental measurements are only capable of measuring product ratios as a function of precursor internal energy. Thus, experimental scientists require such an RRKM analysis to provide a dynamic and atomistic picture of the chemical reaction. The following code determines and plots the final HNC to HCN ratio at internal energies up to 100 000  $\text{cm}^{-1}$ .


```
Do[mechanism_p =
  NDSolve[{
    h-1 VC'[t] ==
      - (k1[[p]] + k2[[p]] + k3[[p]] + k6[[p]] + k9[[p]]) VC[t] +
      k4[[p]] Int1_III[t] + k7[[p]] Int1_IV[t] +
      k10[[p]] Int1_V[t],
    h-1 Int1_III'[t] ==
      k3[[p]] VC[t] - (k4[[p]] + k5[[p]] + k12[[p]] + k13[[p]])
      Int1_III[t],
    h-1 Int1_IV'[t] ==
      k6[[p]] VC[t] - (k7[[p]] + k8[[p]]) Int1_IV[t],
    h-1 Int1_V'[t] == k9[[p]] VC[t] -
      (k10[[p]] + k11[[p]]) Int1_V[t],
    h-1 HCN'[t] == (k1[[p]] + k2[[p]]) VC[t] +
      k5[[p]] Int1_III[t],
    h-1 HNC'[t] == k8[[p]] Int1_IV[t] + k11[[p]] Int1_V[t] +
      (k12[[p]] + k13[[p]]) Int1_III[t],
    h-1 HCCH'[t] == (k1[[p]] + k2[[p]]) VC[t] +
      (k5[[p]] + k12[[p]] + k13[[p]]) Int1_III[t] +
      k8[[p]] Int1_IV[t] + k11[[p]] Int1_V[t],
    VC[0] == 1, Int1_III[0] == 0, Int1_IV[0] == 0, Int1_V[0] == 0,
    HCN[0] == 0, HNC[0] == 0, HCCH[0] == 0
  }, {VC, Int1_III, Int1_IV, Int1_V, HCN, HNC, HCCH}, {t, 1013},
  {p, 21 000, 100 000, 100}]
```

```

productRatios =
  Table[{p, First@Quiet[HNC[1013] / HCN[1013] /. mechanismp]},
    {p, 21 000, 100 000, 1000}] /. {Indeterminate -> 0};

ListLinePlot[
  productRatios,
  AxesOrigin -> {0, 0}, Frame -> True,
  FrameLabel -> {"Energy (cm-1)",
    Row[{"HNC:HCN at t = 1s", Style["t", Italic], " = 1s"]}]
]

```



This determines the energy at which HNC reaches its maximum relative abundance.

```

SortBy[productRatios, Last] // Last

{59 000, 0.707044}

```

The last plot indicates that HCN production is favored at all system internal energies. Iso-cyanide (HNC) only becomes a noticeable product at a system internal energy above 30 000 cm<sup>-1</sup>. This is due to paths IV–VII becoming competitive with path I at higher energies. Surprisingly, the product ratio reduces at energies greater than 58 800 cm<sup>-1</sup> (equivalently, there is a maximum in the product distribution curve). This result would be difficult to predict, even qualitatively, with just the calculated rate constants, demonstrating the usefulness of the powerful tools available in Mathematica.

## ■ References

- [1] D. I. Giacomo, "A Short Account of RRKM Theory of Unimolecular Reactions and of Marcus Theory of Electron Transfer in a Historical Perspective," *Journal of Chemical Education*, **92**(3), 2015 pp. 476–481. doi:10.1021/ed5001312.
- [2] F. A. Lindemann, S. Arrhenious, I. Langmuir, N. R. Dhar, J. Perrin and W. C. McC. Lewis, "Discussion on 'The Radiation Theory of Chemical Action'," *Transactions of the Faraday Society*, **17**, 1922 pp. 598–606. doi: 10.1039/TF9221700598.
- [3] O. K. Rice and H. C. Ramsperger, "Theories of Unimolecular Gas Reactions at Low Pressures," *Journal of the American Chemical Society*, **49**(7), 1927 pp. 1617–1629. doi:10.1021/ja01406a001.
- [4] L. S. Kassel, "Studies in Homogeneous Gas Reactions I," *The Journal of Physical Chemistry*, **32**(2), 1928 pp. 225–242. doi:10.1021/j150284a007.
- [5] R. A. Marcus, "Unimolecular Dissociations and Free Radical Recombination Reactions," *The Journal of Chemical Physics*, **20**(3), 1952 pp. 359–364. doi:10.1063/1.1700424.
- [6] IUPAC. *Compendium of Chemical Terminology*, 2nd ed. (the "Gold Book") (compiled by A. D. McNaught and A. Wilkinson), Blackwell Scientific Publications: Oxford, 1997. goldbook.iupac.org (XML online corrected version, created by M. Nic, J. Jirat and B. Kosata; updates compiled by A. Jenkins).
- [7] T. Beyer and D. F. Swinehart, "Algorithm 448: Number of Multiply-Restricted Partitions," *Communications of the ACM*, **16**(6), 1973 pp. 379. doi.org/10.1145/362248.362275.
- [8] S. E. Stein and B. S. Rabinovitch, "Accurate Evaluation of Internal Energy Level Sums and Densities Including Anharmonic Oscillators and Hindered Rotors," *The Journal of Chemical Physics*, **58**(6), 1973 pp. 2438–2445. doi.org/10.1063/1.1679522.
- [9] Z. Homayoon, S. A. Vázquez, R. Rodríguez-Fernández and E. Martínez-Núñez, "Ab Initio and RRKM Study of the HCN/HNC Elimination Channels from Vinyl Cyanide," *The Journal of Physical Chemistry A*, **115**(6), 2011 pp. 979–985. doi.org/10.1021/jp109843a.
- [10] P. M. Morse, "Diatomic Molecules According to the Wave Mechanics. II. Vibrational Levels," *Physical Review*, **34**(1), 1929 pp. 57–64. doi:10.1103/PhysRev.34.57.

A. C. Mansell, D. J. Kahle and D. J. Bellert, "Calculating RRKM Rate Constants from Vibrational Frequencies and Their Dynamic Interpretation," *The Mathematica Journal*, 2017. dx.doi.org/doi:10.3888/tmj.19-5.

## About the Authors

Adam C. Mansell is a Ph.D. candidate in the Bellert Group of the Department of Chemistry and Biochemistry at Baylor University. He holds a B.A. in chemistry from Concordia University Irvine.

David J. Kahle is a computational statistician and assistant professor of statistics at Baylor University. He holds a B.A. in mathematics from the University of Richmond and M.A. and Ph.D. degrees in statistics from Rice University.

Darrin J. Bellert is an associate professor of chemistry at Baylor University. He received his B.S. from Wright State University in Ohio, and his Ph.D. from Florida State University.

**Adam C. Mansell**

*Department of Chemistry and Biochemistry  
Baylor University  
One Bear Place #97348  
Waco, TX 76798  
adam\_mansell@baylor.edu*

**David J. Kahle**

*Department of Statistical Science  
Baylor University  
One Bear Place #97140  
Waco, TX 76798  
david.kahle@gmail.com*

**Darrin J. Bellert**

*Department of Chemistry and Biochemistry  
Baylor University  
One Bear Place #97348  
Waco, TX 76798  
darrin\_bellert@baylor.edu*